Procurement-Consumer

A. J. Conejo, 2019

THE OHIO STATE UNIVERSITY

What

1. Electricity procurement - Consumer

Procurement-Consumer

Hypotheses:

Procurement via contracts or through the pool

One single contract available throughout the procurement horizon with a fixed price

Uncertain pool prices characterized by scenarios

Procurement-Consumer

Constants:

λ^{C} contract price [\$/MWh]

λ_{ts} pool price at hour t and scenario s [\$/MWh]

α_s weight of scenario s [per unit]

D_t demand to be supplied at hour t [MW]

Procurement-Consumer

Constants:

n_{T} number of time periods

n_{S} number of price scenarios

Procurement-Consumer

Variables

d^{C} hourly demand procured from the contract [MW]

d_t^{P} demand procured from the pool at hour t [MW]

Procurement-Consumer

$$\min_{d^{\mathrm{C}}, d_t^{\mathrm{P}}, \forall t} \qquad c = n_T \, d^{\mathrm{C}} \, \lambda^{\mathrm{C}} + \sum_{s=1}^{n_{\mathrm{S}}} \alpha_s \sum_{t=1}^{n_{\mathrm{T}}} \lambda_{ts} \, d_t^{\mathrm{P}}$$

s.t.

$$d^{\mathrm{C}} + d_t^{\mathrm{P}} = D_t, \quad \forall t$$

Procurement-Consumer

$$\begin{matrix} s=1 & s=2 \\ \frac{1}{4} & \frac{3}{4} \end{matrix}$$

Add to 1

$$[\lambda_{ts}] = \begin{bmatrix} 4 & 6 \\ 5 & 8 \end{bmatrix} \begin{matrix} t=1 \\ t=2 \end{matrix}$$

$$D_t = \begin{bmatrix} 18 \\ 20 \end{bmatrix}$$

$$\lambda^C = \text{between } 6 \text{ \& } 7$$

Procurement-Consumer

$$\min_{d^{\mathrm{C}}, d_1^{\mathrm{P}}, d_1^{\mathrm{P}}} \quad c = 2\, d^{\mathrm{C}}\, \lambda^{\mathrm{C}} + \frac{1}{4}[4d_1^{\mathrm{P}} + 5d_2^{\mathrm{P}}] + \frac{3}{4}[6d_1^{\mathrm{P}} + 8d_2^{\mathrm{P}}]$$

s.t.

$$d^{\mathrm{C}} + d_1^{\mathrm{P}} = 18$$

$$d^{\mathrm{C}} + d_2^{\mathrm{P}} = 20$$

Procurement-Consumer

λ^C [\$/MWh]	6.2	6.3	6.4
d^C [MW]	18	18	0
d_1^P [MW]	0	0	18
d_2^P [MW]	2	2	20

Procurement-Consumer

```
$title EProcurement
variable z;
positive variables dC, dP1, dP2;
scalar d1 /18/, d2 /20/;
scalar lC /6.4/;
scalar l11 /4/, l21 /5/, l12 /6/, l22 /8/;
equations of, s1, s2;
of.. z =e= 2*lC*dC + (1/4)*(l11*dP1+l21*dP2) + (3/4)*(l12*dP1+l22*dP2);
s1.. dC + dP1 =e= d1;
s2.. dC + dP2 =e= d2;
model EPro /all/;
solve EPro using lp minimizing z;
```

Procurement-Consumer

```
                      LOWER     LEVEL     UPPER    MARGINAL

---- VAR z            -INF     244.000    +INF        .
---- VAR dC             .         .       +INF       0.050
---- VAR dP1            .       18.000    +INF        .
---- VAR dP2            .       20.000    +INF        .
```

Procurement-Consumer

	LOWER	LEVEL	UPPER	MARGINAL
---- EQU of	.	.	.	1.000
---- EQU s1	18.000	18.000	18.000	5.500
---- EQU s2	20.000	20.000	20.000	7.250

Procurement-Consumer

What if:

More contracts?

Contracts of different durations?

Deciding on contracts at different time points?

Self-production facility

Risk aversion

Procurement-Consumer: General Formulation

The mathematical formulation of the multi-stage stochastic problem faced by the large consumer is

Procurement-Consumer

Contract

Pool

Self-production

$$\text{Minimize}_{s_{b\omega}, P^{\mathrm{B}}_{b\omega}, E^{\mathrm{P}}_{t\omega}, E^{\mathrm{S}}_{nt\omega}, \zeta, \eta_\omega}$$

$$\sum_{\omega=1}^{N_\Omega} \pi_\omega \sum_{t=1}^{N_\mathrm{T}} \left(\sum_{b \in T_b} \lambda^{\mathrm{B}}_{bt\omega} P^{\mathrm{B}}_{b\omega} d_t + \lambda^{\mathrm{P}}_{t\omega} E^{\mathrm{P}}_{t\omega} + \sum_{n=1}^{N_\mathrm{N}} S^{\mathrm{S}}_n E^{\mathrm{S}}_{nt\omega} \right)$$

$$+ \beta \left(\zeta + \frac{1}{1-\alpha} \sum_{\omega=1}^{N_\Omega} \pi_\omega \eta_\omega \right) \qquad (1)$$

Risk

subject to

Procurement-Consumer

$s_{b\omega}$ binary variable to decide on contracts.

$P^{\mathrm{B}}_{b\omega}$ energy purchased through bilateral contracts.

$E^{\mathrm{P}}_{t\omega}$ energy purchased/sold in the pool

$E^{\mathrm{S}}_{nt\omega}$ energy self-produced.

ζ CVaR auxiliary variable.

η_{ω} CVaR auxiliary variable.

Procurement-Consumer

The objective function consists of two parts. The first term corresponds to the expected procurement. The second term is equal to the CVaR multiplied by a weighting factor β.

The factor β is used to model the tradeoff between the expected cost and the risk of cost variability, measured by the CVaR.

Procurement-Consumer

Solving this problem for different values of β yields a set of solutions with different values of expected cost and CVaR.

Consequently, it is possible to represent the expected cost versus the CVaR or the cost standard deviation for different values of β, thus obtaining the so-called efficient frontier.

Procurement-Consumer

If contract b is selected, the power purchased $P^{\mathrm{B}}_{b\omega}$ is bounded by upper and lower limits $P_b^{\mathrm{B,max}}$ and $P_b^{\mathrm{B,min}}$ as follows:

$$P_b^{\mathrm{B,min}} \ s_{b\omega} \leq P^{\mathrm{B}}_{b\omega} \leq P_b^{\mathrm{B,max}} \ s_{b\omega}, \ \forall b, \forall \omega. \qquad (2)$$

For each contract b with parameters $\{\lambda_b^{\mathrm{B}}, P_b^{\mathrm{B,max}}, P_b^{\mathrm{B,min}}, T_b\}$, the consumer has to decide both whether or not that contract is signed and the level of contracted power.

Procurement-Consumer

$$0 \leq E^{\mathrm{S}}_{1t\omega} \leq \bar{E}^{\mathrm{S}}_{1}, \ \forall t, \forall \omega \tag{3}$$

$$0 \leq E^{\mathrm{S}}_{nt\omega} \leq \bar{E}^{\mathrm{S}}_{n} - \bar{E}^{\mathrm{S}}_{n-1}, \ n = 2, \ldots, N_{\mathrm{N}}, \forall t, \forall \omega \tag{4}$$

Constraints above model the production block bounds of the self-production facility owned by the consumer.

Procurement-Consumer

$$\sum_{n=1}^{N_{\mathrm{N}}} E_{nt\omega}^{\mathrm{S}} + E_{t\omega}^{\mathrm{P}} + \sum_{b \in B_t} P_{b\omega}^{\mathrm{B}} d_t = E_t^{\mathrm{D}} - E_t^{\mathrm{PC}}, \forall t, \forall \omega \qquad (5)$$

Constraints above impose the energy balance for each period and each scenario. B_t is the set of bilateral contracts available in period t, and E_t^{PC} is a constant that represents the quantity of energy previously contracted.

Procurement-Consumer

$$E^{\mathrm{P}}_{t\omega} \geq - \sum_{n=1}^{N_{\mathrm{N}}} E^{\mathrm{S}}_{nt\omega}, \ \forall t, \forall \omega \qquad (6)$$

Constraints above avoid the arbitrage between bilateral contracts and the pool.

Procurement-Consumer

$$P^{\mathrm{B}}_{b\omega} = P^{\mathrm{B}}_{b\omega+1}, \ \forall b, \omega = 1, \ldots, N_{\Omega} - 1, \quad \text{if } A(\omega, K_b) = 1 \tag{7}$$

Constraints above set the non-anticipativity conditions.

Procurement-Consumer

$$\sum_{t=1}^{N_\mathrm{T}} \left(\sum_{b \in B_t} \lambda_{bt\omega}^{\mathrm{B}} P_{b\omega}^{\mathrm{B}} d_t + \lambda_{t\omega}^{\mathrm{P}} E_{t\omega}^{\mathrm{P}} + \sum_{n=1}^{N_\mathrm{N}} S_n^{\mathrm{S}} E_{nt\omega}^{\mathrm{S}} \right) - \zeta \leq \eta_\omega, \ \forall \omega \tag{8}$$

Constraints above are used to compute the CVaR.

Procurement-Consumer

$$\eta_\omega \geq 0, \ \forall \omega \qquad (9)$$

$$s_{b\omega} \in \{0, 1\}, \ \forall b \in B, \forall \omega. \qquad (10)$$

Finally, constraints above constitute variable declarations.

Procurement-Consumer: Example

We consider a planning horizon of four hourly periods, in which contract decisions are made at the beginning of each period.

We consider six contracts: two contracts spanning the four periods, and one contract for each single period. The parameters defining each contract are provided in the table below.

Procurement-Consumer: Example

Contract #	Usage Period	λ_b^{B} (\$/MWh)	$P_b^{\mathrm{B,max}}$ (MW)	$P_b^{\mathrm{B,min}}$ (MW)
1	1-4	48.5	75	20
2	1-4	50.0	75	20
3	1	49.5	50	20
4	2	51.0	50	20
5	3	49.0	50	20
6	4	50.0	50	20

Procurement-Consumer: Example

The electricity demand of the consumer is considered to be deterministic and therefore, defined as independent of pool price scenarios.

The table below shows the electricity demand of the consumer for each period.

Procurement-Consumer: Example

Period #			
1	2	3	4
200	250	225	275

Procurement-Consumer: Example

The pool price is modeled using a set of 16 equiprobable scenarios. The table below provides the pool price for each period and scenario.

Procurement-Consumer: Example

	Period #					Period #			
Scenario #	1	2	3	4	Scenario #	1	2	3	4
1	44	47	45	45	9	50	52	50	51
2	44	47	45	47	10	50	52	50	55
3	44	47	46	47	11	50	52	51	52
4	44	47	46	49	12	50	52	51	53
5	44	44	42	42	13	50	49	47	49
6	44	44	42	45	14	50	49	47	50
7	44	44	44	45	15	50	49	48	50
8	44	44	44	46	16	50	49	48	53

Procurement-Consumer: Example

The corresponding scenario-tree structure is depicted in the figure below:

Procurement-Consumer: Example



Scenario #	Period #				Scenario #	Period #			
	1	2	3	4		1	2	3	4
1	44	47	45	45	9	50	52	50	51
2	44	47	45	47	10	50	52	50	55
3	44	47	46	47	11	50	52	51	52
4	44	47	46	49	12	50	52	51	53
5	44	44	42	42	13	50	49	47	49
6	44	44	42	45	14	50	49	47	50
7	44	44	44	45	15	50	49	48	50
8	44	44	44	46	16	50	49	48	53

Procurement-Consumer: Example

Note that each node is a decision point and each branch represents the realization of a pool price value.

For instance, the two branches leaving the root node represent the two possible realizations of the pool price in period 1, namely, 44 and \$50/MWh.

Procurement-Consumer: Example

The consumer owns a 100-MW self-production unit with a linear production cost equal to \$45/MWh.

Procurement-Consumer: Example

The optimization problem is set up considering a confidence level α equal to 0.95.

The resulting problem includes 480 constraints, 338 real variables, and 96 binary variables.

Procurement-Consumer: Example

A GAMS code to solve this problem is provided at the end of this presentation.

Procurement-Consumer: Example

The optimal solutions in terms of expected cost, cost standard deviation and CVaR are provided in the table below.

Procurement-Consumer: Example

β	Expected cost (\$)	Cost standard deviation (\$)	CVaR (\$)
0	44,073.44	1776.70	46,525.00
1	44,217.97	1340.87	46,012.50
2	44,581.25	908.23	45,750.00
10	44,643.75	843.47	45,737.50

Procurement-Consumer: Example

The solution obtained for $\beta = 0$ attains the lowest expected cost and the highest risk, measured in terms of the standard deviation and the CVaR of the procurement cost.

For $\beta = 10$ the expected cost increases 1.3% to attain a reduction of 52.5% in the standard deviation of the cost and 1.7% in the CVaR.

Procurement-Consumer: Example

The efficient frontier in terms of the expected cost and the cost standard deviation is depicted in the figure below.

Procurement-Consumer: Example





Procurement-Consumer: Example

Low-risk solutions (low standard deviation) are related to high expected costs.

Observe that the solution obtained for $\beta = 1$ is particularly relevant, because it reduces the cost standard deviation by 24.5%, leading to a small increase of 0.3% in the expected cost.

Procurement-Consumer: Example

The amounts of power purchased from bilateral contracts 1 to 3, which are decided at the beginning of the planning horizon, are listed in the table below.

Procurement-Consumer: Example

Contract #	β			
	0	1	2	10
1	0	75	75	75
2	0	0	75	75
3	0	0	0	50

Procurement-Consumer: Example

For $\beta = 0$ no contracts are signed because expected pool prices are smaller than the reference contract prices.

It should be noted that the larger the parameter β, the lower the variability of the cost and the higher the quantity of the power contracted.

These results indicate that the risk associated with the volatility of the procurement cost is efficiently hedged by the use of bilateral contracts.

Procurement-Consumer: Example

The expected electricity procurement is provided in the table below.

Procurement-Consumer: Example

β	Expected contract purchases (MWh)	Expected pool purchases (MWh)	Expected pool sales (MWh)	Expected self-production (MWh)
0	31.25	656.25	0.00	262.50
1	356.25	331.25	0.00	262.50
2	656.25	118.75	87.50	262.50
10	706.25	93.75	112.50	262.50

Procurement-Consumer: Example

The purchases from bilateral contracts grow as the parameter β increases and the risk aversion of the consumer becomes more significant in consequence.

The opposite effect is observed for the pool purchases, i.e., the expected pool purchases decrease as the value of β increases and as a result, the expected pool sales increase with the value of β.

For large values of this parameter, the demand is mostly supplied by contracts, and the energy generated through the self-production unit is sold in the pool if the pool price is higher than the production cost.

Procurement-Consumer: Example

Finally, it is worth mentioning that the self-production remains constant for different values of β. The self-production unit operates in those periods and scenarios where pool prices are higher than the self-production cost.

If the electricity demand of the consumer is not supplied in these periods and scenarios by bilateral contracts, the self-production unit is used to supply part of the electricity demand of the consumer. On the other hand, if the electricity demand is entirely satisfied by bilateral contracts, the electricity obtained from the self-production unit is sold in the pool.

This behavior can be observed in the two tables below.

Consumer example: electricity procurement for $\beta = 0$

Consumer example: electricity procurement for $\beta = 0$

	Period #											
	1			2			3			4		
Scenario #	C	P	SP	C	P	SP	C	P	SP	C	P	SP
1	0	200	0	0	150	100	0	225	0	0	275	0
2	0	200	0	0	150	100	0	225	0	0	175	100
3	0	200	0	0	150	100	0	125	100	0	175	100
4	0	200	0	0	150	100	0	125	100	0	175	100
5	0	200	0	0	250	0	0	225	0	0	275	0
6	0	200	0	0	250	0	0	225	0	0	275	0
7	0	200	0	0	250	0	0	225	0	0	275	0
8	0	200	0	0	250	0	0	225	0	0	175	100
9	0	100	100	0	150	100	50	75	100	50	125	100
10	0	100	100	0	150	100	50	75	100	50	125	100
11	0	100	100	0	150	100	50	75	100	50	125	100
12	0	100	100	0	150	100	50	75	100	50	125	100
13	0	100	100	0	150	100	0	125	100	0	175	100
14	0	100	100	0	150	100	0	125	100	0	175	100
15	0	100	100	0	150	100	0	125	100	50	125	100
16	0	100	100	0	150	100	0	125	100	50	125	100

Procurement-Consumer: Example

The table above provides the energy traded through bilateral contracts and the pool, as well as the self-production in each period and scenario for $\beta = 0$. In this case, neither the first-period nor the second-period contract is signed.

The self-production unit operates in all periods and scenarios where pool prices are higher than the production cost. During period 1, the expected demand (200 MWh) is completely supplied by purchases from the pool in scenarios 1-8.

Procurement-Consumer: Example

The consumer uses solely the pool because the pool price is smaller than the self-production cost. However, in scenarios 9-16, the self-production cost is smaller than the pool price, and the unit is operated then at its full capacity.

A similar behavior can be observed in the remaining periods. In periods 3 and 4, one-period contracts are signed in those scenarios where pool prices are higher than the reference contract prices. This occurs in scenarios 9-12 of period 3, and in scenarios 9-12 and 15-16 of period 4.

Consumer example: electricity procurement for $\beta = 10$

Consumer example: electricity procurement for $\beta = 10$

	Period #											
	1			2			3			4		
Scenario #	C	P	SP	C	P	SP	C	P	SP	C	P	SP
1	200	0	0	150	0	100	150	75	0	150	125	0
2	200	0	0	150	0	100	150	75	0	150	25	100
3	200	0	0	150	0	100	150	-25	100	150	25	100
4	200	0	0	150	0	100	150	-25	100	150	25	100
5	200	0	0	150	100	0	150	75	0	150	125	0
6	200	0	0	150	100	0	150	75	0	150	125	0
7	200	0	0	150	100	0	150	75	0	150	125	0
8	200	0	0	150	100	0	150	75	0	150	25	100
9	200	-100	100	200	-50	100	200	-75	100	200	-25	100
10	200	-100	100	200	-50	100	200	-75	100	200	-25	100
11	200	-100	100	200	-50	100	200	-75	100	200	-25	100
12	200	-100	100	200	-50	100	200	-75	100	200	-25	100
13	200	-100	100	200	-50	100	150	-25	100	150	25	100
14	200	-100	100	200	-50	100	150	-25	100	150	25	100
15	200	-100	100	200	-50	100	150	-25	100	200	-25	100
16	200	-100	100	200	-50	100	150	-25	100	200	-25	100

Procurement-Consumer: Example

Complementarily, the table above provides the purchases of energy in each period and scenario for $\beta = 10$. In this case, contracts 1-3 are signed. For instance, observe that in period 1 the entire consumer demand is supplied through bilateral contracts.

Note that in scenarios 9-16 of period 1 the energy produced by the self-production unit is sold in the pool, which is due to the fact that pool prices are higher than the production cost in these scenarios. A similar behavior is observed in the remaining periods.

GAMS

```
$title MARKET CLEARING UNDER EQUIPMENT FAILURES

***************************************************************************
*        DATA
***************************************************************************

SETS
N    Nodes             /n1 * n3/
I    Units             /i1 * i3/
J    Loads             /j1 * j1/
T    Periods           /t1 * t4/
W    Scenarios         /w0 * w8/
M    Offer blocks      /m1 * m1/
S    System states     /s0*s2/;

ALIAS(N,R);

SCALARS
NT          Number of time periods /4/,
dt          Duration of time periods /1/
x           Reactance of transmission lines /0.13/
Pbase       Power base /41/;
```

GAMS

```
TABLE GDATA(I,*) Generation technical data
         PMIN   PMAX    Istatus    Pini
i1       10     100     0          0
i2       10     100     0          0
i3       10     50      0          0;

TABLE RDATA_G(I,T,*) Generation-side reserve data
                 Rspin_max_up    Rspin_max_dn    Rnspin_max
i1.t1*t4         90              90              100
i2.t1*t4         90              90              100
i3.t1*t4         40              40              50;


TABLE CostBlock(I,T,M) Offer cost of energy blocks
                 m1
i1.t1*t4         30
i2.t1*t4         40
i3.t1*t4         20;

TABLE WidthBlock(I,T,M) Width of energy blocks
                 m1
i1.t1*t4         100
i2.t1*t4         100
i3.t1*t4         50;
```

GAMS

```
TABLE Cr_schG(I,T,*) Generation-side reserve offer cost
                up      dn        NS
i1.t1*t4        5       5         4.5
i2.t1*t4        7       7         5.5
i3.t1*t4        8       8         7;

TABLE ML(J,N) Mapping of the set of loads into the set of
nodes
         n1      n2      n3
j1       0       0       1;

TABLE NetState(N,N,S)   Network states
         s0      s1      s2
n1.n1    0       0       0
n1.n2    1       1       1
n1.n3    1       1       0
n2.n1    1       1       1
n2.n2    0       0       0
n2.n3    1       1       1
n3.n1    1       1       0
n3.n2    1       1       1
n3.n3    0       0       0;
```

GAMS

```
TABLE GenState(I,N,S)   States of unit availability
        s0      s1      s2
i1.n1   1       1       1
i1.n2   0       0       0
i1.n3   0       0       0
i2.n1   0       0       0
i2.n2   1       1       1
i2.n3   0       0       0
i3.n1   0       0       0
i3.n2   0       0       0
i3.n3   1       0       1;

PARAMETERS

elast_limits(J,T,*)   Elasticity limits of loads
/  j1.t1.sup     30
   j1.t2.sup     80
   j1.t3.sup     110
   j1.t4.sup     40
   j1.t1.inf     30
   j1.t2.inf     80
   j1.t3.inf     110
   j1.t4.inf     40 /
```

GAMS

```
RDATA_D(J,T,*) Demand-side reserve data
/  j1.t1.Rspin_max_up     3
   j1.t2.Rspin_max_up     8
   j1.t3.Rspin_max_up     11
   j1.t4.Rspin_max_up     4
   j1.t1.Rspin_max_dn     3
   j1.t2.Rspin_max_dn     8
   j1.t3.Rspin_max_dn     11
   j1.t4.Rspin_max_dn     4 /

prob(W) Scenario w probability
/  w0    0.98807312
   w1    0.00199015
   w2    0.00199015
   w3    0.00199015
   w4    0.00199015
   w5    0.00099157
   w6    0.00099157
   w7    0.00099157
   w8    0.00099157 /
```

GAMS

```
TAU(W) Period in which the contingency defining scenario w occurs
/  w0    5
   w1    1
   w2    2
   w3    3
   w4    4
   w5    1
   w6    2
   w7    3
   w8    4 /

NumO(I,T)        Number of energy blocks offered by each unit
lambdaSU(I,T)    Start-up offer cost
lambdaL(J,T)     Demand utility
Cr_schD(J,T,*)   Demand-side reserve offer cost
Vlol(J,T)        Value of lost load
b(N,N)           Imaginary parts of the admittance of lines
fmax(N,N)        Transmission capacity limits
conec(N,N,T,W)   Network state in time period t and scenario w
MG(I,N,T,W)      Mapping of the set of units into the set of nodes
G(I,T,W)         Set of available units per scenario and period;
```

GAMS

```
NumO(I,T) = 1;
lambdaSU(I,T) = 100;
b(N,R)$(ord(N) ne ord(R)) = -1/x;
G(I,T,W) = 1;
lambdaL(J,T) = 0;
Cr_schD(J,T,'up') = 70;
Cr_schD(J,T,'dn') = 70;
Vlol(J,T) = 1000;
fmax(N,R)$(ord(N) ne ord(R)) = 55;
```

GAMS

```
loop(T,
    loop(W,
         if(ord(T) < TAU(W),
           conec(N,R,T,W) =  NetState(N,R,'s0');
           MG(I,N,T,W) = GenState(I,N,'s0');
         else
           if(ord(W) <= 5,
                conec(N,R,T,W) = NetState(N,R,'s1');
                MG(I,N,T,W) = GenState(I,N,'s1');
                G('i3',T,W) = 0;
           else
                conec(N,R,T,W) = NetState(N,R,'s2');
                MG(I,N,T,W) = GenState(I,N,'s2');
           );
        );
    );
);
```

GAMS

```
***************************************************************************
*        DECLARATION OF VARIABLES
***************************************************************************

VARIABLES
   EC                      Expected cost (objective function value)

** FIRST-STAGE VARIABLES
   Ps(I,T)                 Power output schedule
   Ls(J,T)                 Demand schedule
   R_spin_schG_d(I,T)      Generation-side spinning reserve schedule (downward)
   R_spin_schG_u(I,T)      Generation-side spinning reserve schedule (upward)
   R_spin_schD_d(J,T)      Demand-side spinning reserve schedule (downward)
   R_spin_schD_u(J,T)      Demand-side spinning reserve schedule (upward)
   R_nspin_schG(I,T)       Generation-side non-spinning reserve schedule
   u(I,T)                  Scheduled commitment status
```

GAMS

```
** SECOND-STAGE VARIABLES
   Lc(J,T,W)             Power consumption in real time
   Lshed(J,T,W)          Amount of involuntarily load shed
   Pg(I,T,W)             Generator power output in real time
   pgblock(I,T,W,M)      Power produced from energy blocks
   r_spin_depG_d(I,T,W)  Generation-side spinning reserve deployment (downward)
   r_spin_depG_u(I,T,W)  Generation-side spinning reserve deployment (upward)
   r_spin_depD_d(J,T,W)  Demand-side spinning reserve deployment (downward)
   r_spin_depD_u(J,T,W)  Demand-side spinning reserve deployment (upward)
   r_nspin_depG(I,T,W)   Generation-side non-spinning reserve deployment
   angle(N,T,W)          Voltage angle
   Pinj(N,T,W)           Power injection
   f(N,R,T,W)            Power flow
   v(I,T,W)              Real-time commitment status
   Csu(I,T,W)            Start-up cost;
```

GAMS

```
***************************************************************************
*        MATHEMATICAL CHARACTERIZATION OF VARIABLES
***************************************************************************

POSITIVE VARIABLES
Csu(I,T,W), Lshed(J,T,W), r_spin_depG_d(I,T,W), r_spin_depG_u(I,T,W),
r_nspin_depG(I,T,W), r_spin_depD_d(J,T,W), r_spin_depD_u(J,T,W),
R_spin_schG_d(I,T), R_spin_schG_u(I,T), R_spin_schD_d(J,T), R_spin_schD_u(J,T),
R_nspin_schG(I,T), pgblock(I,T,W,M);


BINARY VARIABLES u(I,T), v(I,T,W);

** Elasticity limits of demand
Ls.up(J,T)= elast_limits(J,T,'sup');
Ls.lo(J,T)= elast_limits(J,T,'inf');

** Reference node
angle.fx('N1',T,W)=0;

** Non-anticipativity constraints
Lshed.fx(J,T,W)$(ord(T) LT TAU(W))= 0;
r_spin_depG_d.fx(I,T,W)$(ord(T) LT TAU(W))= 0;
r_spin_depG_u.fx(I,T,W)$(ord(T) LT TAU(W))= 0;
r_spin_depD_d.fx(J,T,W)$(ord(T) LT TAU(W))= 0;
r_spin_depD_u.fx(J,T,W)$(ord(T) LT TAU(W))= 0;
r_nspin_depG.fx(I,T,W)$(ord(T) LT TAU(W))= 0;
```

GAMS

```
***************************************************************************
*        EQUATIONS
***************************************************************************

EQUATIONS

ECFunction      Objective function

** ELECTRICITY MARKET CONSTRAINTS

** Production limits
MAX_PROD(I,T)
MIN_PROD(I,T)

** Market equilibria
MARKET_EQU(T)

** Scheduled reserve determination constraints

** Generation side
RESERVE_SCH_SPIN_G_u(I,T) Spinning up
RESERVE_SCH_SPIN_G_d(I,T) Spinning down
RESERVE_SCH_NSPIN_G(I,T)  Non-spinning
```

GAMS

```
** Demand side
RESERVE_SCH_SPIN_D_u(J,T) Spinning up
RESERVE_SCH_SPIN_D_d(J,T) Spinning down

** REAL-TIME OPERATING CONSTRAINTS

** Start-up costs
SUo(I,W)          T = 1
SU(I,T,W)         T > 1

** Generation limits
GL1(I,T,W)
GL2(I,T,W)

** Decomposition of generator power outputs into blocks
GL3(I,T,W,M)
GL4(I,T,W)

** Involuntary load shedding constraints
LIMIT_ENS(J,T,W)
```

GAMS

```
** Network constraints
PB(N,T,W)             Power balance
DEF_Pinj(N,T,W)       Definition of power injection
DEF_FLOW(N,R,T,W)     Defintion of power flows
MAX_CAP(N,R,T,W)      Transmission capacity

** LINKING CONSTRAINTS
Pgenerated(I,T,W) Composition of generator power outputs
Pconsumed(J,T,W)  Composition of the power consumption

** Deployed reserve determination constraints

** Generation side
RESERVE_DPL_SPIN_G_u(I,T,W) Spinning up
RESERVE_DPL_SPIN_G_d(I,T,W) Spinning down
RESERVE_DPL_NSPIN_G(I,T,W)  Non-spinning

** Demand side
RESERVE_DPL_SPIN_D_u(J,T,W) Spinning up
RESERVE_DPL_SPIN_D_d(J,T,W) Spinning down;
```

GAMS

```
ECFunction..

EC =e=  dt*sum((I,T),
Cr_schG(I,T,'up')* R_spin_schG_u(I,T)+ Cr_schG(I,T,'dn')* R_spin_schG_d(I,T)
+ Cr_schG(I,T,'NS')* R_nspin_schG(I,T)) + dt*sum((J,T),
Cr_schD(J,T,'up')* R_spin_schD_u(J,T)+ Cr_schD(J,T,'dn')* R_spin_schD_d(J,T))
+ dt*sum((T,W),
prob(W)*(sum((I,M)$(ord(M) LE NumO(I,T)), CostBlock(I,T,M)* pgblock(I,T,W,M))
- sum(J, lambdaL(J,T)* Lc(J,T,W))
+ sum(J, Vlol(J,T)*Lshed(J,T,W))))
+ sum((I,T,W),prob(W)*Csu(I,T,W));

MIN_PROD(I,T).. Ps(I,T) =g= GDATA(I,'PMIN')* u(I,T);

MAX_PROD(I,T).. Ps(I,T) =l= GDATA(I,'PMAX')*u(I,T);

MARKET_EQU(T).. sum(I, Ps(I,T)) =e= sum(J, Ls(J,T));

RESERVE_SCH_SPIN_G_u(I,T)..

                 R_spin_schG_u(I,T) =l= RDATA_G(I,T,'Rspin_max_up')*u(I,T);

RESERVE_SCH_SPIN_G_d(I,T)..

                 R_spin_schG_d(I,T) =l= RDATA_G(I,T,'Rspin_max_dn')*u(I,T);
```

GAMS

```
RESERVE_SCH_NSPIN_G(I,T)..

                R_nspin_schG(I,T) =l= RDATA_G(I,T,'Rnspin_max')*(1-u(I,T));

RESERVE_SCH_SPIN_D_u(J,T).. R_spin_schD_u(J,T) =l= RDATA_D(J,T,'Rspin_max_up');

RESERVE_SCH_SPIN_D_d(J,T).. R_spin_schD_d(J,T) =l= RDATA_D(J,T,'Rspin_max_dn');

SUo(I,W).. Csu(I,'t1',W) =g= lambdaSU(I,'t1')*(v(I,'t1',W)-GDATA(I,'Istatus'));

SU(I,T,W)$(ord(T) GT 1).. Csu(I,T,W) =g= lambdaSU(I,T)*(v(I,T,W)-v(I,T-1,W));

GL1(I,T,W).. Pg(I,T,W) =g= GDATA(I,'PMIN')* v(I,T,W);

GL2(I,T,W).. Pg(I,T,W) =l= GDATA(I,'PMAX')*v(I,T,W);

GL3(I,T,W,M)$(ord(M) LE NumO(I,T)).. pgblock(I,T,W,M) =l= WidthBlock(I,T,M);

GL4(I,T,W).. Pg(I,T,W) =e= sum(M$(ord(M) LE NumO(I,T)), pgblock(I,T,W,M));

LIMIT_ENS(J,T,W).. Lshed(J,T,W)=l= Lc(J,T,W);

PB(N,T,W).. sum(I$(MG(I,N,T,W) EQ 1),Pg(I,T,W))- sum(J$(ML(J,N) EQ 1),
                                Lc(J,T,W)-Lshed(J,T,W))-Pinj(N,T,W) =e= 0;

DEF_Pinj(N,T,W).. Pinj(N,T,W) =e= sum(R$(conec(N,R,T,W) EQ 1),f(N,R,T,W));
```

GAMS

```
DEF_FLOW(N,R,T,W)$(conec(N,R,T,W) EQ 1)..

         f(N,R,T,W)=e= -b(N,R)*(angle(N,T,W)-angle(R,T,W));

MAX_CAP(N,R,T,W)$(conec(N,R,T,W) EQ 1).. f(N,R,T,W) =l= fmax(N,R);

Pgenerated(I,T,W)$(G(I,T,W) EQ 1)..

     Pg(I,T,W) =e= Ps(I,T)+ r_spin_depG_u(I,T,W)+r_nspin_depG(I,T,W)
                  -r_spin_depG_d(I,T,W);

Pconsumed(J,T,W)..

     Lc(J,T,W) =e= Ls(J,T)- r_spin_depD_u(J,T,W) + r_spin_depD_d(J,T,W);

RESERVE_DPL_SPIN_G_u(I,T,W).. r_spin_depG_u(I,T,W) =l= R_spin_schG_u(I,T);

RESERVE_DPL_SPIN_G_d(I,T,W).. r_spin_depG_d(I,T,W) =l= R_spin_schG_d(I,T);

RESERVE_DPL_NSPIN_G(I,T,W).. r_nspin_depG(I,T,W) =l= R_nspin_schG(I,T);

RESERVE_DPL_SPIN_D_u(J,T,W).. r_spin_depD_u(J,T,W)=l= R_spin_schD_u(J,T);

RESERVE_DPL_SPIN_D_d(J,T,W).. r_spin_depD_d(J,T,W)=l= R_spin_schD_d(J,T);
```

GAMS

```
*****************************************************************************
*        MODEL
*****************************************************************************

MODEL MCU /ALL/;
MCU.optcr=0;
Option iterlim = 1e8;
Option reslim = 1e10;
Option mip=cplex;
SOLVE MCU USING mip MINIMIZING EC;
```

Procurement-Consumer

M. Carrion, A. B. Philpott, A. J. Conejo and J. M. Arroyo, “A Stochastic Programming Approach to Electric Energy Procurement for Large Consumers,” in IEEE Transactions on Power Systems, vol. 22, no. 2, pp. 744-754, May 2007.

This is it!